Extracting and uploading HD4DP2 data
This guide is designed for data providers to show how to efficiently extract data from your local HD4DP2 from the previous collection year and seamlessly adapt it for the upcoming one. By following these steps, you can significantly reduce the manual effort involved.
Within these pages, we will detail who is authorized to perform this task, how to extract the data successfully, clarify and provide clear instructions on how to prepare your extracted data for a successful upload in the new collection year. Finally, we will provide instructions on how to upload the prepared datafile. 
Your careful attention to this guide will ensure a smooth and efficient transition.

1. Authorization   
Only the “DP managers” have the access to extract data from the HD4DP2 registry, see roles description https://docs.healthdata.be/documentation/hd4dp2-self-service-portal/roles-and-permissions
Usually, this means only one person/email address will have the authorization to perform the process below. In case you don’t know who your DP manager is, please contact BCFR@sciensano.be.

2. Extraction of the data    
The procedure to extract data from a registry from the local HD4DP2 is described in docs.healthdata.be:
https://docs.healthdata.be/documentation/hd4dp2-self-service-portal/export-csv-files

Be aware to download the correct version (only extract data of the last collection year). 



3. Preparing the extracted data file   
The next step is to guide you through a simple process using LibreOffice, a free and open-source office software similar to Microsoft Office Excel (https://www.libreoffice.org/download/download-libreoffice/). Please note that these instructions were created using LibreOffice with its language set to English. Therefore, some terms may not directly correspond to those in your localized version.

Our goal is to:
· Open a CSV file that you downloaded from HD4DP2 (Healthdata.be).
· Copy a ready-made macro from a Word document into LibreOffice.
· Run the macro to process the data.
You don't need any programming knowledge to follow these steps. Just go step by step and you'll get it done!
Note: depending on the HD4DP2 environment you are working in, the author name, author group and co-author should be compatible. 

How to Open a CSV File in LibreOffice
Once you've downloaded your file from HD4DP2, and you know where it is saved on your computer (e.g., in your Downloads folder), follow these steps:
1. Open LibreOffice Calc.
2. In LibreOffice's main window, click on “File” > “Open”".
3. A file browser window will pop up. Use it to find your CSV file (the one from HD4DP2):
· Look in the folder where you saved it (e.g., "Downloads").
· Click on the file name to select it.
· Then click "Open".
4. A dialog window titled "Text Import" will appear.
· Normally, everything is automatically set to import the data. 
· click "OK".
· If you’ve imported and none of the columns have data, please try to adjust the settings of the dialog window. It is normal that many columns are empty, please don’t adjust any settings before you’ve pressed OK and checked the entire file to see whether data actually didn’t load successfully. 
Your file will now open in LibreOffice. 



How to Add and Run a Macro
Now that your file is open, you need to copy and paste the macro code from the Word document into LibreOffice and run it.
Step-by-step:
1. Below in this document, you will find all the code you need to copy paste into LibreOffice.
2. Select all the code below, then copy it (Ctrl+C or right mouse click + “copy”).
3. Go back to LibreOffice, where your CSV file is open.
4. Click on the Tools (or Extra/Outils for Dutch/French) menu at the top, then choose Macros > Edit Macros
5. A window will appear. In the left panel, click Standard  module 1
6. Once you’ve opened the module 1, remove all the text (Ctrl+A, delete).
7. Paste all the copied code (copied from the bottom of this document; CTRL+V or right mouse button - paste). If this doesn’t work, please try and copy the code again at the bottom of this document. 
8. If everything went well, you should see that the code you pasted should have +- 210 lines (see screenshot below)
[image: ]


9. Next, scroll all the way to the top of the code, and CLICK on the first line. 
[image: ]

10. Press the green play button to run the macro. 
[image: ]
11. Now, many windows will pop up with all the steps that the macro has executed. 

12. Press OK and walk through all windows. If you reach the final window, you can proceed. [image: ]
13. Go back to the LibreOffice “excel” sheet (Green icon on task bar) [image: ]

14. Save the file In the directory you want to save it in, make sure it has a distinct name so you don’t upload the unedited file later on – for example “Stable_data_file”. 
[image: ]
15. Please make sure to “Use Text CSV Format” to save the file! 
This is essential for the next steps.
[image: ]
16. [bookmark: _Hlk199768750]If you have used LibreOffice before, it might be that the CSV separator is not set to the semicolon-separated “;” values format. However, this is a requirement for the upload to HD4DP to successfully execute. 

If you can successfully run the macro but the upload shows an error, please check the saved text CSV file by opening it in Notepad and checking whether columns are separated using a semicolon. 

[image: ]
If other symbols appear as a separator, please adjust your settings using Tools > Options > LibreOffice Calc > Formula > Separators.
[image: ] 

Tips and Troubleshooting
· If you get a security warning about macros, LibreOffice may block them by default. You can allow them by going to:
· Tools > Options > LibreOffice > Security > Macro Security
· Set it to "Medium" so you can choose to allow macros each time.
· If the macro doesn't seem to work, make sure you copied all the code from the Word document correctly.
· If you can successfully run the macro but the upload shows an error, please check the saved text CSV file by opening it in Notepad and checking whether columns are separated using a semicolon. 

[image: ]
If other symbols appear as a separator, please adjust your settings using Tools > Options > LibreOffice Calc > Formula > Separators.
· Please contact nathan.vanhumbeeck@sciensano.be in case of further questions.

Uploading 
The procedure for the upload of data is described in docs.healthdata.be
https://docs.healthdata.be/documentation/belgian-cystic-fibrosis-registry/csv-upload-data-submission-bcfr-registry


Macro code 
(Select everything below this line with your mouse, and copy the text – CTRL+C): 

Sub Click_Double_here_To_Run_The_Full_Macro()
    Dim oDoc As Object
    Dim oSheet As Object
    Dim lastRow As Long
    Dim colIndex_IDC_PAT As Integer
    Dim colIndex_CD_PAT_STA As Integer
    Dim colIndex_NR_YR_COLCN As Integer
    Dim colIndex_FL_PAT_DECEA As Integer
    Dim colIndex_STATUS As Integer
    Dim i As Long
    Dim header As String
    Dim cellValue As Variant ' Use Variant for cellValue to handle different types
    Dim refHeaders() As String
    Dim totalCols As Integer
    Dim keepColumn As Boolean
    Dim oCellRange As Object
    Dim oCell As Object

    oDoc = ThisComponent
    oSheet = oDoc.Sheets(0) ' Get the first sheet

    ' --- SECTION 1: Find Column Indices and Get Last Used Row (Consolidated from GetLastUsedRow_By_IDC_PAT and column finding loops) ---

    ' Define the array of reference headers directly (moved from ClearUnlistedColumns_UsingHeaders)
    refHeaders = Array( _
        "CD_BONE_DNSTY_TEST", "CD_CFR_MODUL_THERPY", "CD_CFR_MODUL_THERPY_CHOICE", "CD_CHROM_1", _
        "CD_CHROM_1_NEW_NOMEN", "CD_CHROM_1_PROTE_NAM", "CD_CHROM_2", "CD_CHROM_2_NEW_NOMEN", _
        "CD_CHROM_2_PROTE_NAM", "CD_CNTRY_ORIG_FATH", "CD_CNTRY_ORIG_MOTH", "CD_DIAGS_CONFRM", _
        "CD_DISAB_DECIS_PLC", "CD_DISAB_DURA", "CD_DISAB_RECOGN", "CD_HSHLD_COMPO", _
        "CD_INFORM_CNSN_SIGNED", "CD_MECON_TRTMNT", "CD_NASL_TEPD", "CD_NEONAT_SCRNG_TEST", _
        "CD_PANCRC_ENZ", "CD_PAT_CNTRY", "CD_PAT_PLC_RESDC", "CD_PAT_SEX", "CD_PAT_STA", "CD_PS", _
        "CD_RECEIVD_LUNG_TRSP", "CD_RECEIVD_TRSP", "CD_SIBL_ALV_WTH_CF", "CD_SIBL_DECEA_WTH_CF", _
        "CD_SIBL_INCL_PAT", "CD_SWEAT_TEST_CL", "CD_SWEAT_TEST_NA", "CD_SYMP", "CD_TPE_SWEAT_TEST", _
        "CD_T_STA_1", "CD_T_STA_1_NEW_NOMEN", "CD_T_STA_2", "CD_T_STA_2_NEW_NOMEN", "CD_WRK_DISAB", _
        "DT_BEST_TRSPLT_LUNG1", "DT_BEST_TRSPLT_LUNG2", "DT_CFTR_OTH_START", "DT_CLIN_DIAGS", _
        "DT_CTR_START", "DT_ELEXTEZIVA_START", "DT_GENOTYPE", "DT_IVA_START", "DT_LAST_GENTPE", _
        "DT_LUMIVA_START", "DT_MOST_RECNT_DEXA", "DT_PAT_DOB", "DT_RENAL_FUNCN_1M_PSTTRPL", _
        "DT_SWEAT_PRE_ELExTEZIVA", "DT_SWEAT_PRE_IVA", "DT_SWEAT_PRE_LUMIVA", "DT_SWEAT_PRE_TEZIVA", _
        "DT_SWEAT_TEST", "DT_TEPD", "DT_TEZIVA_START", "FL_HGHT_FATH_MS", "FL_HGHT_MOTH_MS", _
        "FL_PAT_DECEA", "IDC_PAT", "MS_BEST_TRSPLT_LUNG1_FEV1", "MS_BEST_TRSPLT_LUNG1_FVC", _
        "MS_BEST_TRSPLT_LUNG1_HGHT", "MS_BEST_TRSPLT_LUNG1_WGHT", "MS_BEST_TRSPLT_LUNG2_FEV1", _
        "MS_BEST_TRSPLT_LUNG2_FVC", "MS_BEST_TRSPLT_LUNG2_HGHT", "MS_BEST_TRSPLT_LUNG2_WGHT", _
        "MS_DISAB_RECOGN_YR", "MS_HGHT_FATH", "MS_HGHT_MOTH", "MS_INDEPNDC_PNT", "MS_MOST_RECNT_DEXA", _
        "MS_ORD_FAM", "MS_PILAR1", "MS_RENAL_FUNCN_1M_CREAN", "MS_SWEAT_PRE_ELExTEZIVA", _
        "MS_SWEAT_PRE_IVA", "MS_SWEAT_PRE_LUMIVA", "MS_SWEAT_PRE_TEZIVA", "MS_TOT_PNT", "NR_YR_COLCN", _
        "STATUS", "TX_AUTHOR", "TX_AUTHOR_GR", "TX_CFR_MODUL_THERPY_SPECIF", "TX_COAUTHOR", "TX_CHROM_1_OTH", _
        "TX_CHROM_1_OTH_NEW_NOMEN", "TX_CHROM_1_OTH_PROTE_NAM", "TX_CHROM_2_OTH", "TX_CHROM_2_OTH_NEW_NOMEN", _
        "TX_CHROM_2_OTH_PROTE_NAM", "TX_PAT", "TX_PAT_FIRST_NAM", "TX_PAT_LAST_NAM", "TX_SYMP_OTH_SPECIF", _
        "TX_TTL_BRTHDY_BIOL_CHLD|0|CD_CHLD_BIRTHM", "TX_TTL_BRTHDY_BIOL_CHLD|0|CD_CHLD_BIRTHY", _
        "TX_TTL_BRTHDY_BIOL_CHLD|1|CD_CHLD_BIRTHM", "TX_TTL_BRTHDY_BIOL_CHLD|1|CD_CHLD_BIRTHY", _
        "TX_TTL_BRTHDY_BIOL_CHLD|2|CD_CHLD_BIRTHM", "TX_TTL_BRTHDY_BIOL_CHLD|2|CD_CHLD_BIRTHY", _
        "TX_TTL_BRTHDY_BIOL_CHLD|3|CD_CHLD_BIRTHM", "TX_TTL_BRTHDY_BIOL_CHLD|3|CD_CHLD_BIRTHY", _
        "TX_TTL_BRTHDY_BIOL_CHLD|4|CD_CHLD_BIRTHM", "TX_TTL_BRTHDY_BIOL_CHLD|4|CD_CHLD_BIRTHY", _
        "TX_TTL_BRTHDY_BIOL_CHLD|5|CD_CHLD_BIRTHM", "TX_TTL_BRTHDY_BIOL_CHLD|5|CD_CHLD_BIRTHY", _
        "TX_TTL_BRTHDY_BIOL_CHLD|6|CD_CHLD_BIRTHM", "TX_TTL_BRTHDY_BIOL_CHLD|6|CD_CHLD_BIRTHY", _
        "TX_TTL_BRTHDY_BIOL_CHLD|7|CD_CHLD_BIRTHM", "TX_TTL_BRTHDY_BIOL_CHLD|7|CD_CHLD_BIRTHY", _
        "TX_TTL_BRTHDY_BIOL_CHLD|8|CD_CHLD_BIRTHM", "TX_TTL_BRTHDY_BIOL_CHLD|8|CD_CHLD_BIRTHY", _
        "TX_TTL_BRTHDY_BIOL_CHLD|9|CD_CHLD_BIRTHM", "TX_TTL_BRTHDY_BIOL_CHLD|9|CD_CHLD_BIRTHY", _
        "TX_TTL_HPIN|0|CD_RIZIV_TREAT_PHYS", "TX_TTL_HPIN|1|CD_RIZIV_TREAT_PHYS", _
        "TX_TTL_HPIN|3|CD_RIZIV_TREAT_PHYS", "TX_TTL_HPIN|4|CD_RIZIV_TREAT_PHYS", _
        "TX_TTL_HPIN|5|CD_RIZIV_TREAT_PHYS", "TX_TTL_HPIN|6|CD_RIZIV_TREAT_PHYS", _
        "TX_TTL_TRSP|0|CD_TPE_TRSPLT", "TX_TTL_TRSP|0|CD_TRSPLT_STA", "TX_TTL_TRSP|0|DT_MOST_RECNT_TRSPLT", _
        "TX_TTL_TRSP|0|DT_TRSPLT_WAIT_LIST", "TX_TTL_TRSP|0|NR_YR_TRSPLT", _        
        "TX_TTL_TRSP|1|CD_TPE_TRSPLT", "TX_TTL_TRSP|1|CD_TRSPLT_STA", "TX_TTL_TRSP|1|DT_MOST_RECNT_TRSPLT", _
        "TX_TTL_TRSP|1|DT_TRSPLT_WAIT_LIST", "TX_TTL_TRSP|1|NR_YR_TRSPLT", _        
        "TX_TTL_TRSP|2|CD_TPE_TRSPLT", "TX_TTL_TRSP|2|CD_TRSPLT_STA", "TX_TTL_TRSP|2|DT_MOST_RECNT_TRSPLT", _
        "TX_TTL_TRSP|2|DT_TRSPLT_WAIT_LIST", "TX_TTL_TRSP|2|NR_YR_TRSPLT", _        
        "TX_TTL_TRSP|3|CD_TPE_TRSPLT", "TX_TTL_TRSP|3|CD_TRSPLT_STA", "TX_TTL_TRSP|3|DT_MOST_RECNT_TRSPLT", _
        "TX_TTL_TRSP|3|DT_TRSPLT_WAIT_LIST", "TX_TTL_TRSP|3|NR_YR_TRSPLT", _        
        "TX_TTL_TRSP|4|CD_TPE_TRSPLT", "TX_TTL_TRSP|4|CD_TRSPLT_STA", "TX_TTL_TRSP|4|DT_MOST_RECNT_TRSPLT", _
        "TX_TTL_TRSP|4|DT_TRSPLT_WAIT_LIST", "TX_TTL_TRSP|4|NR_YR_TRSPLT", _        
        "TX_TTL_TRSP|5|CD_TPE_TRSPLT", "TX_TTL_TRSP|5|CD_TRSPLT_STA", "TX_TTL_TRSP|5|DT_MOST_RECNT_TRSPLT", _
        "TX_TTL_TRSP|5|DT_TRSPLT_WAIT_LIST", "TX_TTL_TRSP|5|NR_YR_TRSPLT", _
        "TX_IDC_PAT_INT", "FL_IDC_PAT_GENER", "TX_LANG" _
    )

    ' Find IDC_PAT column index and last used row
    colIndex_IDC_PAT = -1
    For i = 0 To oSheet.Columns.Count - 1
        If Trim(oSheet.getCellByPosition(i, 0).String) = "IDC_PAT" Then
            colIndex_IDC_PAT = i
            Exit For
        End If
    Next i

    If colIndex_IDC_PAT = -1 Then
        MsgBox """IDC_PAT"" column not found! No Data was found. Did you manage to open the HD4DP2 file in Libreoffice? If you can't solve this issue, contact nathan.vanhumbeeck@sciensano.be", 16, "Error"
        Exit Sub
    End If

    lastRow = 0
    For i = 1 To oSheet.Rows.Count - 1
        If Trim(oSheet.getCellByPosition(colIndex_IDC_PAT, i).String) <> "" Then
            lastRow = i
        Else
            Exit For
        End If
    Next i

    If lastRow = 0 Then
        MsgBox "No Data was found. Did you manage to open the HD4DP2 file in Libreoffice? If you can't solve this issue, contact nathan.vanhumbeeck@sciensano.be", 16, "Error"
        Exit Sub
    End If

    ' Find other required column indices (moved from individual subs)
    colIndex_CD_PAT_STA = -1
    colIndex_NR_YR_COLCN = -1
    colIndex_FL_PAT_DECEA = -1
    colIndex_STATUS = -1

    totalCols = oSheet.Columns.Count
    For i = 0 To totalCols - 1
        header = Trim(oSheet.getCellByPosition(i, 0).String)
        Select Case header
            Case "CD_PAT_STA": colIndex_CD_PAT_STA = i
            Case "NR_YR_COLCN": colIndex_NR_YR_COLCN = i
            Case "FL_PAT_DECEA": colIndex_FL_PAT_DECEA = i
            Case "STATUS": colIndex_STATUS = i
        End Select
    Next i

    ' --- SECTION 2: Clear Unlisted Columns (Logic from ClearUnlistedColumns_UsingHeaders) ---

    totalCols = 0
    Do While Trim(oSheet.getCellByPosition(totalCols, 0).String) <> ""
        totalCols = totalCols + 1
    Loop

    For iCol = 0 To totalCols - 1 ' Loop forwards as we are not removing columns
        header = Trim(oSheet.getCellByPosition(iCol, 0).String)
        keepColumn = False
        For i = 0 To UBound(refHeaders)
            If header = refHeaders(i) Then
                keepColumn = True
                Exit For
            End If
        Next i

        If Not keepColumn Then
            ' Clear cells in this column from row 1 down to lastRow
            oCellRange = oSheet.getCellRangeByPosition(iCol, 1, iCol, lastRow)
            oCellRange.clearContents(com.sun.star.sheet.CellFlags.VALUE + _
                                     com.sun.star.sheet.CellFlags.FORMULA + _
                                     com.sun.star.sheet.CellFlags.STRING)
        End If
    Next iCol

    MsgBox "Successfully cleared cells in unlisted columns.", 64, "Done"

    ' --- SECTION 3: Update CD_PAT_STA Column (Logic from Update_CD_PAT_STA_Column) ---

    If colIndex_CD_PAT_STA = -1 Then
        MsgBox """CD_PAT_STA"" column not found! Skipping update.", 16, "Warning"
    Else
        For i = lastRow To 1 Step -1 ' Loop from bottom to top
            cellValue = oSheet.getCellByPosition(colIndex_CD_PAT_STA, i).Value
            Select Case cellValue
                Case 5, 7, 9, 10
                    oSheet.Rows.removeByIndex(i, 1)
                Case 1
                    oSheet.getCellByPosition(colIndex_CD_PAT_STA, i).Value = 2
            End Select
        Next i
        MsgBox "Successfully edited patient status, and removed patients who are not in follow-up.", 64, "Done"
    End If

    ' --- SECTION 4: Increment NR_YR_COLCN (Logic from Increment_NR_YR_COLCN_By_One) ---

    If colIndex_NR_YR_COLCN = -1 Then
        MsgBox """NR_YR_COLCN"" column not found! Skipping increment.", 16, "Warning"
    Else
        For i = 1 To lastRow
            oCell = oSheet.getCellByPosition(colIndex_NR_YR_COLCN, i)
            If oCell.Type = com.sun.star.table.CellContentType.VALUE Then
                oCell.Value = oCell.Value + 1
            End If
        Next i
        MsgBox "Updated the collection year to the new data collection.", 64, "Done"
    End If

    ' --- SECTION 5: Remove Rows with FL_PAT_DECEA = TRUE (Logic from RemoveRowsWith_FL_PAT_DECEA_TRUE) ---

    If colIndex_FL_PAT_DECEA = -1 Then
        MsgBox """FL_PAT_DECEA"" column not found! Skipping row removal.", 16, "Warning"
    Else
        For i = lastRow To 1 Step -1 ' Loop bottom to top to safely remove rows
            cellValue = Trim(UCase(oSheet.getCellByPosition(colIndex_FL_PAT_DECEA, i).String))
            If cellValue = "TRUE" Then
                oSheet.Rows.removeByIndex(i, 1)
            End If
        Next i
        MsgBox "Removed data rows that shouldn't be included.", 64, "Done"
    End If

    ' --- SECTION 6: Set STATUS to "draft" (Logic from Set_STATUS_To_Draft_IfNotEmpty) ---

    If colIndex_STATUS = -1 Then
        MsgBox """STATUS"" column not found! Skipping status update.", 16, "Warning"
    Else
        For i = 1 To lastRow
            oCell = oSheet.getCellByPosition(colIndex_STATUS, i)
            If Trim(oCell.String) <> "" Then
                oCell.String = "draft"
            End If
        Next i
        MsgBox "Set the status to DRAFT for the new collection year.", 64, "Done"
    End If

    ' --- SECTION 7: Replace Semicolons with Commas ---
    MsgBox "Starting replacement of semicolons with commas...", 64, "Processing"

    ' Re-determine totalCols as rows might have been removed in previous sections,
    ' and we need the actual last column after column clearing in Section 2.
    ' This ensures we only process relevant columns.
    totalCols = 0
    Do While Trim(oSheet.getCellByPosition(totalCols, 0).String) <> ""
        totalCols = totalCols + 1
    Loop
    totalCols = totalCols - 1 ' Adjust to be the actual last column index (0-based)

    ' Re-determine lastRow as rows might have been removed in previous sections.
    lastRow = 0
    For i = 1 To oSheet.Rows.Count - 1
        If Trim(oSheet.getCellByPosition(colIndex_IDC_PAT, i).String) <> "" Then
            lastRow = i
        Else
            Exit For
        End If
    Next i

    If lastRow = 0 Then ' Check if any data rows remain after previous operations
        MsgBox "No data rows remaining to process for semicolon replacement.", 48, "Warning"
    Else
        For i = 1 To lastRow ' Loop through rows (starting from 1 to skip header)
            For j = 0 To totalCols ' Loop through columns
                oCell = oSheet.getCellByPosition(j, i)
                ' Perform replacement directly on the cell's string value
                If InStr(oCell.String, ";") > 0 Then ' Check if semicolon exists
                    oCell.String = Replace(oCell.String, ";", ",")
                End If
            Next j
        Next i
        MsgBox "Successfully replaced all semicolons with commas.", 64, "Done"
    End If

    MsgBox "Awesome, everything worked! You are ready to save this file and upload the stable data :)", 64, "Done"
End Sub

